
ABSTRACT  

Starting with a SAS PLOT program, we will transfer this plot into PROC GPLOT and review the many 
ways you can improve the look of the plot using SAS GRAPH statements. We will make the plot really 
shine by customizing titles, footnotes, symbols, legends, axes, and even the reference line. At each step, 
a hands-on example will be presented where the user will choose their own features such as symbol 
colors and placement of the legend. In the end, you will have built your own personalized graph using the 
Title, Footnote, Symbol, Legend, and Axis statements. 

INTRODUCTION  

The data to be used in the plot is a SAS dataset that contains daily counts of hits on each of 3 websites 
for one month.  The SAS dataset is sorted by date and has the following variables. 

 Date – format yymmdd8. 

 Day – day of month in $char2. format 

 Web1 – hits on Website 1 

 Web2 – hits on Website 2 

 Web3 – hits on Website 3 

 Total – total number of hits on all three websites 
 

We are given a graph produced using a Proc Plot program and asked to improve the look of the graph.  
We start with the Proc Plot graph and take a look at the graph it produces. 

 

PROC PLOT PROGRAM 

This is the initial Proc Plot program that we started with: 

DATA perm.hits; 
 SET perm.hits; 
 Day=substr (date,7,2); 
 Label Day=’Day of Month in August’; 
 Label Total=’Total Number of Hits’; 
 Label web1=’Number of Hits’; 
RUN; 
TITLE ‘Number of Hits on Websites 1, 2, and 3’; 
TITLE2 ‘For the Month of August 2007’; 
TITLE3 ‘Figure 8’; 
FOOTNOTE ‘Company Name  9/15/07’; 
PROC PLOT DATA=perm.hits; 

PLOT  Web1 * day = ‘*’ 
 Web2 * day = ‘+’ 
 Web3 * day = ‘-‘ 
 / OVERLAY  HREF = ‘17’  BOX; 

RUN; 

PROC PLOT GRAPH 

 
                     Number of Hits on Websites 1, 2, and 3    

                          For the Month of August 2007 

                                    Figure 8 

 

                     Plot of web1*day.  Symbol used is '*'. 

                     Plot of web2*day.  Symbol used is '+'. 

                     Plot of web3*day.  Symbol used is '-'. 

 

          „--+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+--† 

   N      |                                  |                              | 

   u 1500 +                                  - - -                          + 

   m      |                                  |     -     -                  | 

   b      |                          * * * * * *     - -   - - -   - - -    | 

   e      |  *   *     *     * * * *         |   *               -       -  | 

   r 1000 +    *   * *   * *                 |     * *                      + 

          |                                  |                              | 

   o      |    +                   +   +     |         *                    | 

   f      |  +   + + + + + + + + +   +   + + + + + + +   * + + + + + + + +  | 

      500 +                                  |             * *              + 

   H      |                              - - |                 * *          | 

   i      |      - - - - -         - - -     |                     * *   *  | 

   t      |  - -           - - - -           |                         *    | 

   s    0 +                                  |                              + 

          |                                  |                              | 

          +--+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+--Œ 

             0 0 0 0 0 0 0 0 0 1 1 1 1 1 1 1 1 1 1 2 2 2 2 2 2 2 2 2 2 3 3 

             1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0 1 

 

                                Day of Month in August 

 

NOTE: 2 obs hidden. 

 

                              Company Name  9/15/07 

 

 

STEPS TO IMPROVE THE PROC PLOT GRAPH WITH PROC GPLOT 

STEP 1 – COMPARING PLOT AND GPLOT 

A review of the options used in the PLOT statement in the PROC PLOT program determined that the 
BOX option is not available for use in GPLOT, but the FRAME option is and they do very similar things.  
Note: another option that is available to GPLOT and not PLOT is the GRID option, which causes 
reference lines to be drawn at every major tick mark.  However, a full grid on this plot would make the plot 
much too busy, so this option was not used.  The OVERLAY and HREF options are valid in both PLOT 
and GPLOT, so those options were kept.  Below is the code used to run the PROC GPLOT program with 
the change to the options on the PLOT statement from BOX to FRAME.  Example code is below: 

 

TITLE ‘Number of Hits on Websites 1, 2, and 3’; 
TITLE2 ‘For the Month of August 2007’; 
TITLE3 ‘Figure 9’; 
PROC GPLOT DATA=perm.hits; 

PLOT Web1 * day = ‘*’ 
 Web2 * day = ‘+’ 
 Web3 * day = ‘-‘ 
 / OVERLAY  HREF = ‘17’  FRAME; 

RUN; 

 

Quick note on an alternative to the OVERLAY option, you can also get a similar effect with three variables 
by using the statement PLOT Y*X=Z.  This will automatically turn on the LEGEND (off by default in 
GPLOT) and will plot one line for each value of Z. 

STEP 2 – MODIFYING THE TITLES AND FOOTNOTES 

 

PROC GPLOT allows changes to font, height, and color of the text.  PROC GPLOT also allows the 
placement of the title to be altered.  Using PROC GPLOT, a box can be drawn around the title or the title 
can be underlined.  Note these options are only available for certain SAS Graph procedures and are not 
available with PROC PLOT.  All of these options work for text in titles, notes, and footnotes. 

 

OPTION WHAT IT DOES 

COLOR = Color of the text 

BCOLOR = Background color if using the box option 

FONT = Font of the text 

HEIGHT = Height of the letters 

JUSTIFY = Left | Center | Right Justifies portions of the text 

ANGLE = degrees Turns the entire line of text the number of degrees specified 

ROTATE =degrees Turns each letter in the text the number of degrees specified 

BOX = 1 | 2 | 3 | 4 Creates a box around the text – 1 lightest line, 4 = darkest line 

DRAW = (coordinate pairs) Draws lines between the pairs of points 

UNDERLINE = 0 | 1 | 2 | 3 Underlines text (0 = no underline, 3 = darkest underline) 

 

All of these options take effect on the text AFTER the option appears.  So if we want the first part of the 
line left justified and the second part right justified, use a command like this: 

 TITLE JUSTIFY=left  ‘first part’  JUSTIFY=right  ‘second part’;   

In our example below, the first change uses the FONT= option to specify a standard font for the entire title 
(by default, TITLE1 uses Swiss font and the others use the default hardware font).  The second change, 
uses the HEIGHT= option to make the first title the same hieght as the other titles.  In order to clearly 
label the month for this plot, the words ‘August 2007’ in the second line of the title were highlighted in red.  
Note that when the title is split in two sections, you must embed the spaces between the sections.  SAS 
will not automatically put a space between them.  (See the extra space after the word ‘of’ in TITLE2 
below). 

A modification of the footnote was added to separate the two parts of the text.  The name of the company 
was left justified and the date was right justified using the JUSTIFY= option.  Example code is below: 

 

TITLE FONT=’Times New Roman’ HEIGHT=1.5  
‘Number of Hits on Websites 1, 2, and 3’; 

TITLE2 FONT=’Times New Roman’ HEIGHT=1.5   
 ‘For the Month of ‘  COLOR=red ‘August 2007’; 
TITLE3 FONT=’Times New Roman’ HEIGHT=1.5 ‘Figure 10’; 
FOOTNOTE JUSTIFY=left ‘Company Name’ 
 JUSTIFY=right ‘September 15, 2007’; 
PROC GPLOT DATA=perm.hits; 

PLOT Web1 * day = ‘*’ 
 Web2 * day = ‘+’ 
 Web3 * day = ‘-‘ 
 / OVERLAY  HREF = ‘17’  FRAME; 

RUN; 

 

 

 

STEP 3 – ADJUSTING SYMBOLS 

PROC GPLOT allows symbols to be specified a couple of ways.  The first is similar to the method used in 
PROC PLOT.  However, the ‘*’ symbol produces a different symbol in PROC GPLOT as compared to 
PROC PLOT.  In order to use an asterisk, you must specify the “=’star’” on the plot statement.  Note that 
there is no way to use a dash as the symbol in PROC GPLOT.  If you specify =’dash’, you will get a circle 
with a dot in the middle (included here as an example).  See Appendix A for a table of the different 
symbols available.  Example code is below: 

PROC GPLOT DATA=perm.hits; 
PLOT Web1 * day = ‘star’ 
 Web2 * day = ‘plus’ 
 Web3 * day = ‘-‘ 
 / OVERLAY  HREF = ‘17’  FRAME; 

RUN; 

 

 

 

The SYMBOL Statement 

Another way to specify the symbol is through the use of the SYMBOL statement.  The symbol statement 
is a very powerful tool providing many other functions besides selecting a symbol.  Using the symbol 
statement, the symbols on the plot can be connected, the types of lines used on the plot can be chosen 
and the colors of the symbols and lines can be selected.  Here are some of the options you can use: 

 

OPTION WHAT IT DOES 

COLOR =  To specify color (also can use CI, CV and CO for colors of the line, points, and 
outlines, respectively. 

VALUE =  Selects the symbol printed on the plot (see Appendix A for the special symbols 
available) 

HEIGHT = To specify size of each symbol on the plot 

INTERPOL  
 
      = BOX 
      = HILO 
      = JOIN 
      = NONE 
      others 

Specifies if you want a line drawn and what type of line to use (also used to ask for 
regression plots) 
      For box plots 
      To draw a single line between min and max values of one axis 
      Connect the dots type of line 
      No line (default) 
      See SAS documentation for others 

LINE = Specifies the line to use (see Appendix B) 

WIDTH = Specifies the thickness of the line 

POINTLABEL = Labels the plot points 

Symbol statements are numbered 1 through 99 and are used consecutively for each combination of plot 
variables.  In this example, we have three separate plots (web1*day, web2*day and web3*day).  If the 
symbol statement does not specify a color, then the SAS system uses that symbol statement repeatedly 
through all the colors in the color palette before going to the next symbol statement.  It is therefore 
recommended that color be specified in each symbol statement so that the user is not just relying only on 
the symbol color to determine the meaning of the plot line or symbol.  Color can be specified for both the 
line and the symbol (COLOR=), or can be specified separately for each (CL=line color and CV=symbol 
color).  For this example, three symbol statements, each with a different color, were used.    

 

To select the symbols, the VALUE= (V=) option is used.  Refer to the SAS documentation for a table of 
the symbols available.  To choose the size of the symbol, use the HEIGHT= (H=) option.  Default height is 
1.  Example code is below: 

 

SYMBOL1 COLOR=blue VALUE=dot H=1; 
SYMBOL2 COLOR=red VALUE=star H=1; 
Symbol3 COLOR=green VALUE=circle H=1; 
PROC GPLOT DATA=perm.hits; 

PLOT  Web1*day  Web2*day  Web3*day  
 / OVERLAY  HREF = ‘17’ FRAME; 

      RUN; 

 

 

 

 

To specify that the points be connected, use the INTERPOL= (or I=) option.  This feature offers many 
variations on connecting the symbols.  For example, to use a straight “connect the dots” type of line, use 
‘I=join’, or to get a regression line use ‘I=R’, or to get a step line use ‘I=needle’.  If you do not wish to 
connect the symbols, choose I=none.  For this plot, a straight line between the points makes the plot 
much easier to understand, so ‘I=join’ was used.   

To select what type of line to use, use the LINE= (L=) option.  See Appendix B for a list of the lines that 
may be used.  There are 46 different lines that can be selected.  Line Type 1 is a solid line and Line Type 
0 is no line.  All the others are combinations of dashed lines of different sized dashes.  To choose the 
thickness of the line use the Width= (W=) option.   The default is 1.  Example code is below: 

 

SYMBOL1 COLOR=blue INTERPOL=join LINE=1 VALUE=dot; 
SYMBOL2 COLOR=red INTERPOL=join LINE=2 VALUE=star; 
SYMBOL3 COLOR=green INTERPOL=join L=3 VALUE=circle; 
PROC GPLOT DATA=perm.hits; 

PLOT  Web1*day  Web2*day  Web3*day  
 / OVERLAY  HREF = ‘17’ FRAME; 

      RUN; 

 

 

 

 

STEP 4 – ADJUSTING THE LEGEND 

Using overlay with the PROC GPLOT procedure does not automatically produce a legend.  To produce a 
legend, use the LEGEND option in the Plot statement.  Example code is below: 

 

PROC GPLOT DATA=perm.hits; 
PLOT Web1*day  Web2*day  Web3*day  
 / OVERLAY  HREF = ‘17’ FRAME LEGEND; 

RUN; 

The LEGEND Statement 

 

Note that in the plot above, the label for Web1 displays in the legend rather than the variable name.  It is 
possible to specify your own legend labels using the LEGENDx statement and the VALUE= option.  In the 
LEGENDx statement, the ‘x’ is a number from 1 to 99, which labels the different legend statements.  
Within the VALUE= option you can specify your own text, justify it, and/or specify different portions of the 
text different colors, fonts, or height. To change the name of the legend itself (in the example above, the 
default label was ‘PLOT’), we can use the LABEL= option. 

 

LEGEND statements specify the characteristics of the legend, but do not create the legend.  The 
characteristics that can be specified include the position and appearance of the legend box, the text and 
appearance of the legend label, the appearance of the legend entries, including the size and shape of the 
legend values, and the text of the labels for the legend values. 

 

OPTION WHAT IT DOES 

LABEL = (text edit options) Modify/specify the legend title 

VALUE = (text edit options) Modify the legend value descriptions 

SHAPE = BAR | LINE | SYMBOL Specifies the size and shape of the legend values 

POSITION =  
       (  <Bottom | Middle | Top> 
           <Left | Center | Right> 
           <Outside | Inside >  ) 

Positions the legend on the graph.   
By default, POSITION=(Bottom Center Outside) 

MODE = Protect | Reserve | 
Share 

Defines how the legend interacts with the graphing area. 

ORDER =  Orders the legend values 

ACROSS =         
DOWN = 

Specifies how legend entries will appear 

FRAME   Draws a frame around the legend 

FWIDTH = Specifies the thickness of the frame 

OFFSET = (x,y) Moves the legend 

CBLOCK =, CSHADOW =, 
CBORDER =, CFRAME 

Specifies color of various parts of the legend or to get shadow 
effects. 

 

Text Description Suboptions can be used whenever you want to modify the appearance of text (inside the 
label and value options, for example).  Text description suboptions affect all the strings that follow unless 
the suboption is changed or turned off. 

 

OPTION WHAT IT DOES 

COLOR =  Specifies color 

FONT =  Specifies font 

HEIGHT = Specifies size of font 

JUSTIFY = LEFT | CENTER | RIGHT Justifies portions of the text 

POSITION =  
      ( <Bottom | Middle |Top > 
        <Left | Center | Right >   ) 

Places the legend label in relation to the legend entries.  
Used only with the LABEL option.   
By default, POSITION=LEFT.   

TICK = Modifies one legend engtry 

 

Unlike the symbol statements, which are automatically used as the plotting process needs them, you will 
need to explicitly specify which legend statement to use for your plot.  To do this, specify legend=legendx 
(where x is which legend statement you are referring to) on the plot statement.  Example code is below: 

 

LEGEND1 LABEL=(HEIGHT=1 ‘Web Sites’) 
   VALUE=(‘Web Site 1’ ‘Web Site 2’ ‘Web Site 3’); 
PROC GPLOT DATA=perm.hits; 

PLOT  Web1*day  Web2*day  Web3*day  
 / OVERLAY  HREF = ‘17’ FRAME LEGEND=legend1; 

RUN; 

In addition to specifying the text for the legend entries, the entries may be repositioned within the legend 
itself.  Using the ACROSS= and DOWN= options, I specified that the legend should be listed vertically 
rather than horizontally. 

 

The legend may also be moved either inside or outside (default) the plotting area using the POSITION= 
option.  If the legend is placed outside the plot, this will limit how much plotting space is actually given to 
the plot.  If you wish to save plotting space, move the legend inside the plotting area.  The default position 
is (Bottom Center Outside).  I chose to move it to (Top Left Inside).   

 

If the legend is moved to the inside of the plotting area, the option MODE must be changed to either 
‘PROTECT’ or ‘SHARE’.  With Protect, the legend covers any points that would otherwise appear in the 
same place as the legend.  With Share, the legend and the points are both printed (on top of each other).  
For this example, PROTECT was used. 

 

To move the legend label to the top of the list of legend entries, use the LABEL= option and its suboptions 
POSITION and JUSTIFY to move it above the entries and center it.  Example code is below: 

 

LEGEND1 LABEL=(HEIGHT=1 POSITION=top JUSTIFY=center 
    ‘Web Sites’) 
  VALUE=(‘Web Site 1’ ‘Web Site 2’ ‘Web Site 3’) 
  ACROSS=1 DOWN=3 
  POSITION= (top left inside) 
  MODE=protect; 
PROC GPLOT DATA=perm.hits; 

PLOT  Web1*day  Web2*day  Web3*day  
 / OVERLAY  HREF = ‘17’ FRAME LEGEND=legend1; 

RUN; 

Two additional options to make the legend stand out are FRAME, to add a frame around the legend and 
CFRAME to shade it.  These two options are mutually exclusive in GPLOT.  To add a frame, use the 
FRAME option in the legend statement.  To shade the legend, use the CFRAME option and specify the 
color you want.  I preferred the CFRAME option for this example. 

 

In order to improve the placement of the legend, it was moved slightly to the right rather than against the 
vertical axis.  To move the legend to the right, use the option OFFSET=(x,y).  The x specifies the amount 
of space to move horizontally, and y specifies the amount of space to move the legend vertically.  With 
the OFFSET option, you can specify the units you want used for the x and y values.  I chose to use 
percent of the plotting space (pct).  You can also specify to use inches (in), or centimeters (cm).   If the x 
or y value is not specified, it is assumed to be zero.  Example code is below: 

 

LEGEND1 LABEL=(HEIGHT=1 POSITION=top JUSTIFY=center 
    ‘Web Sites’) 
  VALUE=(‘Web Site 1’ ‘Web Site 2’ ‘Web Site 3’) 
  ACROSS=1  DOWN=3 
  POSITION = (top left inside)     
  MODE=protect 
  OFFSET = (3 pct)       
  CFRAME = yellow; 
PROC GPLOT DATA=perm.hits; 

PLOT  Web1*day  Web2*day  Web3*day  
 / OVERLAY  HREF = ‘17’ FRAME LEGEND=legend1; 

RUN; 

STEP 5 – ADJUSTING THE AXES 

 

The axes can be modified in two ways.  You may use the options on the PLOT statement in the GPLOT 
procedure or you may use the AXIS statement and then assign the particular AXIS definition in the PLOT 
statement.  

 

In this example, modifying the vertical axis so there are fewer numbers displayed will make the plot look 
cleaner and easier to read.  To do this use the VAXIS= option.  Since these numbers were numeric it is 
possible to specify the axis using a range.  During a test run, I noticed SAS placed four tick marks 
between each major tick mark.  I thought that three would look better and be easier to understand (to 
represent every 50 hits).  To do this, use the VMINOR= option.  Example code is below: 

 

PROC GPLOT DATA=perm.hits; 
PLOT  Web1*day  Web2*day  Web3*day  
 / OVERLAY  HREF = ‘17’ FRAME LEGEND=legend1 
 VAXIS=0 to 1800 by 200   VMINOR=3; 

Run; 

 

 

 

 

The AXIS Statement 

 

The second way to modify the axes is to use the AXIS statement.  Here we are using it to modify the 
placement of the axis labels (not available in the options on the plot statement).   The AXIS statement is 
similar to the LEGEND and SYMBOL statements.  You can specify up to 99 different AXIS statements by 
specifying AXISx where x can be any number from 1 to 99.  To choose which axis definition to use, use 
the VAXIS= and/or the HAXIS= options on the PLOT statement.  This will replace the range of numbers in 
the previous example. 

AXIS statements specify the characteristics of an axis, including the way the axis is scaled, how the data 
values are ordered, the location and appearance of the axis lines and tick marks, and the text and 
appearance of the axis label and major tick marks.  Many of the options that can be used in the AXIS 
statement are the same as the TITLE statement options (i.e., ANGLE and ROTATE suboptions). 

 

OPTION WHAT IT DOES 

COLOR = Color of the axes 

LENGTH = Specify axis length 

STYLE = Specify line type to use for the axis 

WIDTH = Specify the thickness of the axis line 

MAJOR = (tick mark suboptions) Specify the major tick marks 

MINOR = (tick mark suboptions) Specify the minor tick marks 

ORDER =  Value list 

LABEL = (text arguments) Specify the label on the axis 

REFLABEL = (text arguments) Add a label to a reference line on the plot 

VALUE = (text arguments) Specify replacement values for the tick marks 

 
Use the ANGLE and ROTATE suboptions in the LABEL option to change the angle of the vertical axis 
label to display it down the side instead of across the top.  Note the suboptions on the AXIS statement are 
applied to any text that comes after the options.  If you put the text before the suboptions, the suboptions 
will be ineffective.  The ANGLE suboption causes the entire label to be turned.  The ROTATE suboption 
will turn each letter separately.  Note: the direction of –90 degrees will turn the same amount as 
specifying 270 degrees.  The size of the label was increased using the HEIGHT suboption in the label 
option. 

 

In this example, two changes were made to the horizontal axis.  First, a label was added to the reference 
line, using the REFLABEL= option.  Second, removal of the leading zeroes from the tick mark labels for 
the 1st through the 9th of August using the VALUE= option.  Note that these values are character, so the 
text must be specified in quotes and a range could not be used (as is possible with numeric variables).  
The SAS system assigns the values from the first point consecutively for as many labels as provided, and 
then uses the default label for the rest of the points.  Example code is below: 

 

AXIS1  LABEL=(ANGLE=270 ROTATE=90 HEIGHT=1.5  
‘Number of Hits’) 

  ORDER=(0 to 1800 by 200) 
  MINOR=(NUMBER=3) 
  ; 
AXIS2 REFLABEL=(POSITION=top JUSTIFY=center ‘Email Ad‘) 
  VALUE=(‘1’ ‘2’ ‘3’ ‘4’ ‘5’ ‘6’ ‘7’ ‘8’ ‘9’); 
PROC GPLOT DATA=perm.hits; 

PLOT  Web1*day  Web2*day  Web3*day  
 / OVERLAY  HREF = ‘17’ FRAME LEGEND=legend1 
 VAXIS=axis1 HAXIS=axis2; 

RUN; 

GOPTIONS STATEMENT 

The GOPTIONS statement can help you improve your graph even further as well as preparing the graph 
to be saved in a file.  There are over 125 options to choose from – only a few will be described here.  
Some of the options will set global values of font selection and text appearance, others affect color, and 
yet others affect the overall size and positioning of the graph. 

 

OPTION NAME DESCRIPTION 

RESET To reset all GOPTIONS back to the default values 

DEVICE / TARGET Specifies what file type the graph should be created in 

GSFNAME To associate a graph to a file set up with a FILENAME statement. 

GSFMODE REPLACE or APPEND to the output file 

HPOS / VPOS Specify number of columns and rows, respectively, in the graphics output 
area does not affect the width or height of the graphics area, but merely 
divides it into cells. 

FTEXT / HTEXT / CTEXT Font, height, and color for the text in the graph, respectively 

FTITLE / HTITLE /CTITLE Font, height, and color for the title, respectively 

FBY / HBY / CBY Font, height, and color for the BY-line 

HSIZE / VSIZE Specify size of graph in inches or centimeters 

XMAX / YMAX Specify maximum size of graph in inches or centimeters 

HORIGIN / VORIGIN The horizontal and vertical offset of the graph  

BORDER Will draw a border around the graphics output area 

SAVING A PLOT TO A FILE 

 

The first use of the GOPTIONS statement that we will take a look at is to save a plot to a file.  There are 
several options used for this.  To get a list of the possible DEVICE types that can be used, run the 
following code (there are very many devices available): 

 

 

 

 

 

Following is one example that uses some of the options listed above to create a graph in a file: 

 

 

 

 

 

 

  

 

CREATING MULTIPLE GRAPHS IN ONE FILE USING ODS 

 

Using ODS, we can create a single file that includes multiple graphs.  Output files can be in HTML, RTF, 
PDF, and Printer (and others as well) .  Most devices can be used, however, some of the more common 
ones are:  GIF, JPEG, PNG, ActiveX, SASPRTC and EMF. 

 

Here is one example of how to produce a PDF file with multiple graphs: 

 

 

 

 

 

 

 

  

FILENAME  OUTPLOT ‘C:\PLOT.GIF’; 
GOPTIONS DEVICE=GIF570 
  GSFNAME=OUTPLOT 
  GSFMODE=REPLACE 
  HPOS=83 
  VPOS=43 
  ; 
PROC GPLOT …; 
 …. 
RUN; 

PROC GDEVICE  
CATALOG=sashelp.devices NOFS; 

 LIST; 
RUN; QUIT; 

ODS PDF File = “c:\graphs.pdf"; 
GOPTIONS   

DEVICE  = SASPRTC      
TARGET  = SASPRTC 

          FTEXT   = 'helvetica'         
FTITLE  = 'helvetica/bold'; 

 
 … … PROC GPLOT code (that produces many graphs) … … 
 
ODS PDF CLOSE; 

ODS AND CHANGING GRAPH ATTRIBUTES 

 

You can modify the graph’s overall size by using the XMAX and YMAX options.  These values affect the 
final size of the graphics display area, ergo, the size of the graph.  This is a nice way to control how large 
the file is that is placed into your file.  Following is one example of how this is done: 

 

 

 

 

 

 

 

 

 

 

CHANGING PAGE LAYOUT 

 

GOPTIONS are not the only SAS statement you can use to improve how your graph is stored in the 
output file, there are also some very useful options in the SAS OPTIONS statement: 

 

OPTION NAME DESCRIPTION 

ORIENTATION Landscape or Portrait 

TOPMARGIN Sets the top margin in the output file 

BOTTOMMARGIN Sets the bottom margin in the output file 

LEFTMARGIN Sets the left margin in the output file 

RIGHTMARGIN Sets the right margin in the output file 

 

 

Following is an example using these options: 

 

 

 

 

 

 

 

 

 

GOPTIONS  
RESET=all  
DEVICE=png  
TARGET=png  
XMAX=7 in  
YMAX=9 in  
FTEXT=‘Arial’; 
 

ODS RTF file=‘c:\graph.rtf’; 
SAS Graph statements to create one or many graphs. 

ODS RTF close; 

OPTIONS  
ORIENTATION=landscape  
TOPMARGIN=1.5  
BOTTOMMARGIN=1  
LEFTMARGIN=1  
RIGHTMARGIN=1; 

GOPTIONS  
RESET=all  
DEVICE=sasprtc  
BORDER; 

 
ODS PRINTER SAS PRINTER=‘PostScript EPS Color’  FILE=‘c:\graph.eps’; 
 …SAS Graph statements to create one or more graphs… 
ODS printer close; 

FINAL PROC GPLOT PROGRAM 

 

With only one plot to produce, we decided to create a single GIF file as output.  However, it could easily 
have also been placed in an RTF or PDF file using ODS.  Our final program that we used is below: 

 

       FILENAME  OUTPLOT ‘C:\PLOT.GIF’; 
       GOPTIONS DEVICE=GIF570 
  GSFNAME=OUTPLOT 
  GSFMODE=REPLACE 
  HPOS=83 
  VPOS=43 
                          ; 

LEGEND1 LABEL=(HEIGHT=1 POSITION=top JUSTIFY=center 
    ‘Web Sites’) 
  VALUE=(‘Web Site 1’ ‘Web Site 2’ ‘Web Site 3’) 
  ACROSS=1  DOWN=3 MODE=protect 
  POSITION = (top left inside)     
  OFFSET = (3 pct)  CFRAME = yellow; 
AXIS1  LABEL=(angle=270 ROTATE=90 HEIGHT=1.5  

‘Number of Hits’) 
  ORDER=(0 to 1800 by 200) 
  MINOR=(number=3); 
AXIS2 REFLABEL=(POSITION=top JUSTIFY=center ‘Email Ad‘) 
  VALUE=(‘1’ ‘2’ ‘3’ ‘4’ ‘5’ ‘6’ ‘7’ ‘8’ ‘9’); 
SYMBOL1 COLOR=blue INTERPOL=join LINE=1 VALUE=dot; 
SYMBOL2 COLOR=red INTERPOL=join LINE=2 VALUE=star; 
SYMBOL3 COLOR=green INTERPOL=join LINE=3 VALUE=circle; 
TITLE FONT=’Times New Roman’ HEIGHT=1.5  

‘Number of Hits on Websites 1, 2, and 3’; 
TITLE2 FONT=’Times New Roman’ HEIGHT=1.5   
 ‘For the Month of ‘  COLOR=red ‘August 2007’; 
FOOTNOTE JUSTIFY=left ‘Company Name’ 
 JUSTIFY=right ‘September 15, 2007’; 
PROC GPLOT DATA=perm.hits; 

PLOT  Web1*day  Web2*day  Web3*day  
 / OVERLAY  HREF = ‘17’ FRAME LEGEND=legend1 
 VAXIS=axis1 HAXIS=axis2; 

RUN; 

CONCLUSION 

The options described here are by no means exhaustive of what the SAS GRAPH system can do.  
Please refer to the documentation to learn what else is possible to improve the quality of your graphs.   

 

CONTACT INFORMATION 

Your comments and questions are valued and encouraged. Contact the author at: 

Wendi Wright  
1351 Fishing Creek Valley Rd 
Harrisburg, PA 17112 
(c) 717-421-9315  
wwright@questarai.com 

 
SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of 
SAS Institute Inc. in the USA and other countries. ® indicates USA registration.  

Other brand and product names are trademarks of their respective companies.  

Appendix A 

Special Symbols for Plotting Data Points 

Appendix B 

Line Types 

 

 


